# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=xorg-xmessage
pkgver=1.0.5
pkgrel=1
pkgdesc="Display a message or query in a window"
arch=(x86_64)
url="https://xorg.freedesktop.org/"
license=('custom')
depends=('libxaw')
makedepends=('xorg-util-macros')
source=(https://xorg.freedesktop.org/archive/individual/app/xmessage-${pkgver}.tar.bz2{,.sig})
sha512sums=('a8423c3986b1576fa67800c315a605a527bb9a8f594ac269a736ae7b91eed0d4d83deb89148197dc3cd3f0e51894146958ce2979f78e97dcdb99c5326e34dabe'
            'SKIP')
validpgpkeys=('4A193C06D35E7C670FA4EF0BA2FB9E081F2D130E') # "Alan Coopersmith <alan.coopersmith@oracle.com>"

build() {
  cd xmessage-${pkgver}
  ./configure --prefix=/usr
  make
}

package() {
  cd xmessage-${pkgver}
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}/usr/share/licenses/${pkgname}/COPYING"
}
